

MODULE 3

H₂ molecular emission and collisional-radiative modelling.

Demonstration script

Hugh Summers, Francisco Guzman, Kurt Behringer,
Martin O'Mullane and Alessandra Giunta

October 10, 2013

Contents

1 Demo (a) Looking at molecular data	2
1.1 Demo (a) Figures	4
1.1.1 Demo (a-1) demo_a/demo_a_1_avalue.pdf	4
1.1.2 Demo (a-1) demo_a/demo_a_1_fcondon.pdf	5
1.1.3 Demo (a-2) demo_a/adas901.pdf	6
1.2 Demo (a) Procedures	7
1.2.1 Demo (a-1) demo_a/demo_a.pro	7
2 Demo (b) Rovibronic emission from heteronuclear diatomics	9
2.1 Demo (b) Figures	11
2.1.1 Demo (b-1) demo_b/demo_b_1_BeD_temp.ps	11
2.1.2 Demo (b-1) demo_b/demo_b_1_BeD_BeT.ps	12
2.2 Demo (b) Procedures	13
2.2.1 Demo (b-1) demo_b/demo_b.pro	13

1 Demo (a) Looking at molecular data

DEMO A: Looking at molecular data

PURPOSE: Look at H2 fundamental molecular constants and data.
As for atoms and ions, ADAS includes data format for molecules called mdf (molecular data formats).
This demo shows how to look and extract data from mdf00 and mdf02 data files.

The data files mdf00 contain the following molecular constants, collected in different subdirectories:

- enu: vibrational substate energies for each electronic state;
- pot: electronic potential curve for each electronic state;
- dip: electronic dipole curve for each electronic state;
- aval: vibrationally resolved A-values for each electronic state pair;
- fcf: vibrationally resolved Franck-Condon factors for each electronic state pair.

The data files mdf02 provide the impact process molecular data and are designed to have both the numerical and parameters of the formulaic representation (if available) of each transition.

As for the atomic data, there are routines and interactive programs which deal with the molecular data files.

The idl routine to read and extract data from mdf00 is xxdatm_00.pro. ADAS901 deals interactively with mdf02 files.

EXAMPLE: For this demo, data for H2 molecule are considered. These data are read and extracted using the idl routine xxdatm_00.pro to look at A-values (/home/adas/adas/mdf00/aval/h2/aval0112_0109.dat) and Franck-Condon factors (/home/adas/adas/mdf00/fcf/h2/fc0101_0102.dat).

The input file for the ADAS901 run is /home/adas/adas/mdf02/h2/fg13_h2#e.dat. The process 11 is selected for this example.

DEMO a1: Extract data from mdf00: understanding the molecular data format

1. Use xxdatm_00.pro to read the A-values.
2. Plot a surface of A-values as a function of the lower and upper vibrational levels.
3. Use xxdatm_00.pro to read the Franck-Condon factors.
2. Plot a surface of Franck-Condon factors as a function of the lower and upper vibrational levels.

Program: demo_a.pro

Samples of output: demo_a_1_avalvalue.ps, demo_a_1_fconcondon.ps

DEMO a2: Running ADAS901 interactively

1. Use ADAS901 interactively.
 2. Select the process number.
- Sample of output: adas901.ps

1.1 Demo (a) Figures

1.1.1 Demo (a-1) demo_a/demo_a.1_avalues.pdf




Figure 1: A-value surface from mdf00

1.1.2 Demo (a-1) demo_a/demo_a.1_fcondon.pdf




Figure 2: Franck-Condon factor surface from mdf00

1.1.3 Demo (a-2) demo_a/adas901.pdf

MOLECULAR COEFFICIENT VERSUS ENERGY/TEMPERATURE :

ADAS : ADAS RELEASE: v4.0 PROGRAM: ADAS901 v1.0 DATE: 18/06/13 TIME: 16:01
FILE : /home/giunta/adas/mdf02/h2/fg13_h2#e.dat
TRANSITION : in : 0, 0, ; out : 0, 0, ; process : 11 ; diss ch :
KEY : (CROSSES - INPUT DATA) (FULL LINE - FORMULA FIT)




Figure 3: Cross-section output graph from ADAS901

1.2 Demo (a) Procedures

1.2.1 Demo (a-1) demo_a/demo_a.pro

```
pro demo_a

; Extract A-values for H2 A-value between state 9 and 12

;Define the mdf00 file using aval directory
file = '/home/adas/adas/mdf00/aval/h2/aval0112_0109.dat'

;Use xxdatm_00.pro to read the mdf00/aval selected
xxdatm_00, file=file, fullldata=all_aval

x = indgen(all_aval.max_vib_l+1)
y = indgen(all_aval.max_vib_u+1)

;Plot the A-values as a function of lower and upper vibrational levels
set_plot,'ps'
device, /isolatin1, font_index=8
device, bits=8, filename='demo_a_1_avalvalue.ps', $
      font_size = 14, xsize=18.0, ysize=16.0, $
      yoffset=7.0, /color
device, /helvetica

surface, all_aval.data, x, y, $
      xtitle = 'Lower vibrational levels', $
      ytitle = 'Upper vibrational levels', $
      ztitle = 'Transition probability (/s)'

device,/close
set_plot,'x'

; Frank-Condon factor for

;Define the mdf00 file using fcf directory
file = '/home/adas/adas/mdf00/fcf/h2/fc0101_0102.dat'

;Use xxdatm_00.pro to read the mdf00/fcf selected
xxdatm_00, file=file, fullldata=all_fcf

x = indgen(all_fcf.max_vib_l+1)
y = indgen(all_fcf.max_vib_u+1)

;Plot the Franck-Condon factors as a function of lower and upper
;vibrational levels
set_plot,'ps'
```

```
device, /isolatin1, font_index=8
device, bits=8, filename='demo_a_1_fcondon.ps', $
      font_size = 14, xsize=18.0, ysize=16.0, $
      yoffset=7.0, /color
device, /helvetica

surface, all_fcf.data, x, y, $
      xtitle = 'Lower vibrational levels', $
      ytitle = 'Upper vibrational levels', $
      ztitle = 'Frank-Condon Factor'

device,/close
set_plot,'x'

end
```


2 Demo (b) Rovibronic emission from heteronuclear diatomics

DEMO B: Rovibronic emission from heteronuclear diatomics

PURPOSE: To calculate rovibrational band features for heteronuclear diatomics using the Pickett method.

This demo use as input the mdf01 molecular data format.

The data files mdf01 are grouped in the following subdirectories:

- nist: extract of NIST molecular constant table for selected electronic state;
- behr: driver parameter file for ADAS908 rovibrational spectra analysis (relevant electronic and vibrational transitions from Behringer method for H2)
- pickett: driver parameter file for ADAS908 rovibrational spectra analysis (relevant electronic and vibrational transitions from Pickett method for heteronuclear diatomics)

Different formats are used for the three types of mdf01 and so also different reading routines are used:

- readf_mdf01p.pro for /mdf01/pickett
- readf_mdf01b.pro for /mdf01/behrr

The program to execute the Pickett calculations is wrapper_pickett.pro. This program calls the routine readf_mdf01p.pro to read the mdf01 in the Pickett format.

EXAMPLE: For this example the molecules BeD and BeT are analysed.

The input files are the following:

- /home/adas/adas/mdf01/pickett/bed.dat
- /home/adas/adas/mdf01/pickett/bet.dat

Three rotational and vibrational temperatures are specified for BeD, assuming that rotational temperature, T_r , is equals to vibrational temperature, T_v :

- 1000 K
- 3500 K
- 9000 K

One temperature ($T_r=T_v=3500$ K) has been selected to compare BeD with BeT.

DEMO b1: Calculate rovibrational band features using the Pickett method

1. Specify the wavelength range.
2. Define three different sets of rotational and vibrational temperatures.

3. Use wrapper_pickett.pro to execute the Pickett calculations for BeD at the three sets of temperatures.
 4. Plot the normalised emissivity as a function of pixel.
 5. Compare the emissivity of BeD at the three temperatures.
 6. Use wrapper_pickett.pro to execute the Pickett calculations for BeT at one temperature.
 7. Compare BeD emissivity with BeT emissivity.
- Program: demo_b.pro
Samples of output: demo_b_1_BeD_temp.ps, demo_b_1_BeD_BeT.ps

2.1 Demo (b) Figures

2.1.1 Demo (b-1) demo_b/demo_b.1_BeD_temp.ps




Figure 4: Simulation of the superposed 0-0, 1-1 and 2-2 sequence bands of BeD A2II - $X^2\Sigma$ at three different temperatures, $T_r=T_v=1000$ K, 3500 K and 9000 K.

2.1.2 Demo (b-1) demo_b/demo_b.1_BeD_BeT.ps




Figure 5: Comparison between BeD and BeT bands at $T_r=T_v=3500$ K.

2.2 Demo (b) Procedures

2.2.1 Demo (b-1) demo_b/demo_b.pro

```
pro demo_b
;To calculate rovibrational band features for heteronuclear diatomics
;using the Pickett method
;The program accesses driver datasets of format mdf01/pickett using
;read_mdf01p.pro. The dataset contains a dedicated set of parameters in the
;Pickett format for a specific molecule such as BeD
;The first step is specify the name of the dataset for input

adas_colors, colors=colors

;Specify the input dataset in the mdf01/pickett format
infiled='/home/adas/adas/mdf01/pickett/bed.dat'
infilet='/home/adas/adas/mdf01/pickett/bet.dat'

;Define wavelength range in nm
xmin=485.d0
xmax=505.d0

;Define rotational temperature in K
ktr=3500.d0
ktr_low=1000.d0
ktr_high=9000.d0

;define vibrational temperature in K
ktv=3500.d0
ktv_low=1000.d0
ktv_high=9000.d0

;Use wrapper_pickett.pro to execute the Pickett calculations
;for BeD at three different rotational and vibrational temperatures
wrapper_pickett, infiled, xmin=xmin, xmax=xmax, $
    ktr=ktr_low, ktv=ktv_low,$
    digi_spec=digi_specd_low

wrapper_pickett, infiled, xmin=xmin, xmax=xmax, $
    ktr=ktr, ktv=ktv,$
    digi_spec=digi_specd

wrapper_pickett, infiled, xmin=xmin, xmax=xmax, $
    ktr=ktr_high, ktv=ktv_high,$
    digi_spec=digi_specd_high

;Use wrapper_pickett.pro to execute the Pickett calculations
```

```

;for BeT
wrapper_pickett, infilet, xmin=xmin, xmax=xmax, $
    ktr=ktr, ktv=ktv,$
    digi_spec=digi_spect

!p.thick=1.3
!p.charsize=1.4
!p.charthick=1.3

;Look at the behaviour of BeD as a function of rotational and
;vibrational temperature
;Plot the normalised emissivity spectrum of BeD
;as a function of pixel number for three temperatures

set_plot,'ps'
device, /isolatin1, font_index=8
device, bits=8, filename='demo_b_1_BeD_temp.ps', $
    font_size = 14, xsize=18.0, ysize=16.0, $
    yoffset=7.0, /color
device, /helvetica

plot, digi_specd_low, title='BeD', xtitle=' PIXEL NUMBER', $
    ytitle='NORMALISED EMISSIVITY'
xyouts,100.,0.8,'1000 K'

oplot,digi_specd, color=colors.green
xyouts,100.,0.7,'3500 K',color=colors.green

oplot,digi_specd_high,color=colors.blue
xyouts,100.,0.6,'9000 K',color=colors.blue

device,/close
set_plot,'x'

;Compare BeD and BeT ata specified temperature

set_plot,'ps'
device, /isolatin1, font_index=8
device, bits=8, filename='demo_b_1_BeD_BeT.ps', $
    font_size = 14, xsize=18.0, ysize=16.0, $
    yoffset=7.0, /color
device, /helvetica

plot, digi_specd, xtitle=' PIXEL NUMBER', $
    ytitle='NORMALISED EMISSIVITY'
xyouts,100.,0.8,'BeD'

```

```
oplot, digi_spect, color=colors.red
xyouts,100.,0.7,'BeT', color=colors.red

device,/close
set_plot,'x'

stop
end
```